


Arduino Bluetooth Robot Car Lab Manual

WolfieWeb Advanced Arduino Projects




Build a Bluetooth-controlled robot car using an Arduino Uno, L298N motor driver, HC-05 Bluetooth module, DC motors, and safe power wiring. This manual is designed to be printed and used on the workbench.

Safety first: never power the drive motors directly from the Arduino 5V pin. Use the L298N motor power input and make all grounds common.

1. Parts Checklist


Part	Purpose	Notes
Arduino Uno	Main controller	Runs the robot code and reads Bluetooth commands.
L298N Motor Driver	Motor control	Handles direction and speed control for both DC motors.
HC-05 Bluetooth Module	Wireless control	Pairs with phone Bluetooth serial controller app.
2WD chassis + TT motors	Robot body	Includes wheels, motors, caster, and frame.
Battery pack	Motor power	Use 7.4V to 9V depending on your kit.
Jumper wires	Connections	Use color coding: red power, black ground, yellow signal.
1k and 2k resistors	Voltage divider	Protects HC-05 RX pin from direct 5V logic.

2. Full Wiring Layout



Use the full layout image to see where every major part belongs before you plug in the first wire. Place the Arduino where the USB port is easy to reach. Keep the L298N close to the motors. Keep Bluetooth wires away from motor output wiring when possible.

3. Pinout Reference




Common ground is mandatory. The Arduino, L298N, HC-05, and battery negative must all share ground. Without common ground, signals will behave randomly or fail completely.

4. Pin Mapping Table

Connection	Arduino Pin / Destination	Wire Color	Important Note
L298N ENA	D5	Yellow	PWM speed for left motor
L298N ENB	D6	Yellow	PWM speed for right motor
L298N IN1	D7	Orange	Left motor direction
L298N IN2	D8	Orange	Left motor direction
L298N IN3	D9	Green	Right motor direction
L298N IN4	D12	Green	Right motor direction
HC-05 TXD	D10	Blue	Bluetooth transmit to Arduino receive
HC-05 RXD	D11 through divider	Purple	Use 1k/2k resistor divider
Battery +	L298N +12V/VIN	Red	Motor power only
Battery -	Common GND	Black	Tie to Arduino and module ground


5. Close-Up Wiring: L298N Motor Driver



The L298N is the muscle of the project. The Arduino control pins do not power the motors. They only tell the L298N which direction to spin and how fast to drive each motor.

Check ENA and ENB jumpers: if your board has physical enable jumpers installed, remove them when using Arduino PWM speed control on D5 and D6.

6. Close-Up Wiring: HC-05 Bluetooth Module



The HC-05 TX pin connects to Arduino D10. The Arduino D11 pin sends data to HC-05 RX. Because many HC-05 RX pins prefer 3.3V logic, run Arduino D11 through a simple voltage divider before it reaches HC-05 RX.

7. Step-by-Step Build

1. Mount the TT motors, wheels, and caster wheel on the chassis.
2. Mount the Arduino Uno and L298N so the USB port and screw terminals remain reachable.
3. Connect left and right motors to the L298N motor output terminals.
4. Wire L298N ENA, ENB, IN1, IN2, IN3, and IN4 to the Arduino pins in the table.
5. Wire HC-05 VCC to 5V and GND to common ground.
6. Wire HC-05 TXD to Arduino D10.
7. Wire Arduino D11 to HC-05 RXD through a 1k/2k voltage divider.
8. Connect battery positive to L298N motor power and battery negative to common ground.
9. Upload the code. If upload fails, temporarily disconnect Bluetooth wires from D10 and D11.
10. Pair your phone with HC-05 and send F, B, L, R, and S commands.

8. Bluetooth Command Table

Command	Action
F	Forward
B	Backward
L	Turn left
R	Turn right
S	Stop
0-9	Optional speed control

9. Arduino Code

```
#include <SoftwareSerial.h>

SoftwareSerial bluetooth(10, 11);

const int ENA = 5;
const int ENB = 6;
const int IN1 = 7;
const int IN2 = 8;
const int IN3 = 9;
const int IN4 = 12;

int motorSpeed = 180;

void setup() {
  pinMode(ENA, OUTPUT);
  pinMode(ENB, OUTPUT);
  pinMode(IN1, OUTPUT);
  pinMode(IN2, OUTPUT);
  pinMode(IN3, OUTPUT);
  pinMode(IN4, OUTPUT);
  Serial.begin(9600);
  bluetooth.begin(9600);
  stopCar();
}

void loop() {
  if (bluetooth.available()) {
    char c = bluetooth.read();

    if (c >= '0' && c <= '9') {
      motorSpeed = map(c - '0', 0, 9, 90, 255);
      return;
    }

    if (c == 'F') forward();
    else if (c == 'B') backward();
    else if (c == 'L') left();
    else if (c == 'R') right();
    else stopCar();
  }
}

void setSpeed() {
  analogWrite(ENA, motorSpeed);
  analogWrite(ENB, motorSpeed);
}

void forward() {
  setSpeed();
  digitalWrite(IN1, HIGH);
  digitalWrite(IN2, LOW);
  digitalWrite(IN3, HIGH);
  digitalWrite(IN4, LOW);
}

void backward() {
  setSpeed();
  digitalWrite(IN1, LOW);
  digitalWrite(IN2, HIGH);
  digitalWrite(IN3, LOW);
  digitalWrite(IN4, HIGH);
}

void left() {
  setSpeed();
  digitalWrite(IN1, LOW);
  digitalWrite(IN2, HIGH);
  digitalWrite(IN3, HIGH);
  digitalWrite(IN4, LOW);
}

void right() {
  setSpeed();
  digitalWrite(IN1, HIGH);
  digitalWrite(IN2, LOW);
}
```

```
    digitalWrite(IN3, LOW);  
    digitalWrite(IN4, HIGH);  
}  
  
void stopCar() {  
    digitalWrite(IN1, LOW);  
    digitalWrite(IN2, LOW);  
    digitalWrite(IN3, LOW);  
    digitalWrite(IN4, LOW);  
    analogWrite(ENA, 0);  
    analogWrite(ENB, 0);  
}
```

10. Troubleshooting

Motors do not move: Check battery voltage, L298N power input, ENA/ENB wiring, and common ground.

Bluetooth pairs but commands fail: Confirm 9600 baud, verify the phone app sends single characters, and check TX/RX wiring.

Robot spins in circles: One motor is reversed. Swap that motor pair at the L298N output terminal.

Arduino resets when motors start: The battery is sagging or motors are pulling too much current. Use stronger motor power and keep motor current away from Arduino 5V.

Upload fails: Disconnect HC-05 serial wires from D10/D11 during upload, then reconnect after code is loaded.

Live tutorial path: https://www.wolfieweb.com/arduino_advanced.html